

Paper 160-31

“Best-in-Class” Comparison Group (BCCG) Methodology for Process of Care Performance Indicators

William F. McCarthy, Maryland Medical Research Institute, Baltimore, MD
Nan Guo, Maryland Medical Research Institute, Baltimore, MD

ABSTRACT

This paper presents a methodology for making a meaningful comparison of facilities when assessing performance indicators in health care. A SAS[®] Program with a SAS[®] Macro is presented that reflects the methodology discussed.

Key Words: Performance Indicator Assessment; Best Practice of Care; Outcomes and Effectiveness Research; SAS[®] Program; SAS[®] Macro

INTRODUCTION

The specifications outlined in this document are for process of care performance indicators (the higher the rate the better the compliance with “standard-of-care” practice or evidence-based gold standards). These indicators are measures of the usage of “standard-of-care” or evidence-based gold standards (i.e., the usage of the “best practice of care”).

Performance is measured by the proportion of patients who receive clinically appropriate care when they are eligible for such care [i.e., the number of eligible patients receiving clinically appropriate care divided by the total number of patients who were eligible for such care].

For example, AMI-1: Aspirin at Arrival. Performance is measured by the proportion of AMI patients who receive aspirin within 24 hours before or after hospital arrival who are clinically eligible for receiving aspirin (All AMI patients without aspirin contra-indications).

Historically, a facility’s performance has been compared to the “average” performance of all facilities considered for assessment or some other grouping of facilities (pre-defined “peer-group”). The specifications outlined in this document will provide a methodology for constructing a more meaningful comparison group for assessing how well a facility’s performance is compared to a comparative norm.

The comparison group to be used in this outlined methodology is the “best-in-class” facilities (i.e., a group of facilities considered in the assessment who represent a measurable level of excellence and always exceed average performance). We will call his comparison group the “best-in-class” comparison group (BCCG).

In addition, this methodology ensures that facilities with high performance based on small denominators do not unduly influence how the “best-in-class” comparison group is defined and how its performance (proportion) is defined.

The BCCG can be generated from all the facilities considered in the assessment (project-wide). The BCCG can also be generated from an appropriately defined “peer-group”. Or the BCCG can be generated from some other appropriately defined grouping of facilities.

The BCCG should have at least 10 appropriately selected facilities, each having at least 5 eligible patients for the performance indicator being assessed. Other rules can be applied for the minimum number of facilities needed and the minimum number of eligible patients needed to generate a meaningful BCCG.

BEST-IN-CLASS COMPARISON GROUP (BCCG) METHODOLOGY

The following outline of the BCCG Methodology will use a concrete example for illustration. We will consider AMI-1: Aspirin at Arrival. Performance is measured by the proportion of AMI patients who receive aspirin within 24 hours before or after hospital arrival who are clinically eligible for receiving aspirin (All AMI patients without aspirin contraindications).

Data were collected from 24 different facilities. The raw data are found in Table 1.

Table 1. Raw Data from the 24 Facilities.

Facility #	Eligible AMI Patients	Eligible AMI Patients Receiving Aspirin
1	33	29
2	1	1
3	0	0
4	1	0
5	123	95
6	56	45
7	64	29
8	33	28
9	7	7
10	15	7
11	16	14
12	28	21
13	8	7
14	28	23
15	23	21
16	31	26
17	62	46
18	4	4
19	15	11
20	3	3
21	2	2
22	85	61
23	4	4
24	65	41
<i>TOTAL</i>	<i>707</i>	<i>525</i>

STEP 1.

Determine the number of eligible patients to be used in the “Best-in-Class” Comparison Group.

By the ABC System definition, the top 10 percent of the eligible patient population considered in the assessment (of the 24 facilities) defines the “best-in-class” comparison group. Table 1 indicates that we have 707 eligible patients. Therefore, 10 percent of 707 is 70.7 (approximately 71 of the 707 eligible patients).

Next, we will need to determine which 71 of the 707 eligible patients are the so-called “top” members. Steps 2 through 5 are used for this determination. These “top” members are the patients of facilities giving the “top” level of quality care.

STEP 2.

Calculate the Bayesian Adjusted Performance Fraction (APF) for each facility (row) found in Table 1.

NOTE: This Bayesian APF minimizes the problems associated with computing performance based on small denominators. For example, if a facility had only one eligible patient, and this eligible patient received aspirin (100 percent performance), should this facility be ranked higher than a facility that had 9 out of 10 eligible patients receiving aspirin (90 percent performance)? This is an example of the so-called small denominator problem. This

Bayesian APF reduces the impact of facilities with small numbers of eligible patients on the determination of the "best-in-class" comparison group².

The formula for the Bayesian Adjusted Performance Fraction (APF) is

$APF = (x + 1) / (d + 2)$, where x = the total number of eligible AMI patients receiving aspirin; d = the total number of eligible AMI patients.

For facility # 1, $APF = (29 + 1) / (33 + 2) = 0.857$

STEP 3.

Rank order all facilities in descending order of their APF. Refer to Table 2.

Table 2. Rank Order of Facility APF.

Facility #	Eligible AMI Patients	Eligible AMI Patients Receiving Aspirin	APF
9	7	7	0.889
15	23	21	0.880
1	33	29	0.857
11	16	14	0.833
23	4	4	0.833
18	4	4	0.833
8	33	28	0.829
16	31	26	0.818
13	8	7	0.800
14	28	23	0.800
20	3	3	0.800
6	56	45	0.793
5	123	95	0.768
21	2	2	0.750
17	62	46	0.734
12	28	21	0.733
22	85	61	0.713
19	15	11	0.706
2	1	1	0.667
24	65	41	0.627
3	0	0	0.500
10	15	7	0.471
7	64	29	0.455
4	1	0	0.333
<i>TOTAL</i>	<i>707</i>	<i>525</i>	

STEP 4.

Calculate the cumulative number of eligible patients in the rank ordered table (Table 2). Refer to Table 3.

Table 3. Rank Order of Facility APF with Cumulative Number of Eligible Patients.

Facility #	Eligible AMI Patients	Eligible AMI Patients Receiving Aspirin	APF	Cumulative # of Eligible AMI Patients
9	7	7	0.889	7
15	23	21	0.880	30
1	33	29	0.857	63
11	16	14	0.833	79
23	4	4	0.833	83
18	4	4	0.833	87
8	33	28	0.829	120
16	31	26	0.818	151
13	8	7	0.800	159
14	28	23	0.800	187
20	3	3	0.800	190
6	56	45	0.793	246
5	123	95	0.768	369
21	2	2	0.750	371
17	62	46	0.734	433
12	28	21	0.733	461
22	85	61	0.713	546
19	15	11	0.706	561
2	1	1	0.667	562
24	65	41	0.627	627
3	0	0	0.500	627
10	15	7	0.471	642
7	64	29	0.455	706
4	1	0	0.333	707
TOTAL	707	525		

STEP 5.

Identify the break point at above which 10 percent (approximately 71) of the eligible patients fall.

This break point (also known as the benchmark cut-off) falls at facility # 11 (refer to Table 3).

NOTE: The APF for facility # 11 is the same for facility # 23 and facility # 18. Because of this tie, the break point is moved down to the level of facility # 18 in Table 3. Moving the break point down to the level of facility # 18 gives us 87 eligible patients being above the break point (87 eligible patients being in the “top” level of quality care; the so-called “top” 10 percent). Moving the break point from facility # 11 to facility # 18 resulted in a shift from the top 10 percent to the top 12.3 percent. This is not a problem; it does not dramatically affect the formation of the “best-in-class” comparison group.

The “best-in-class” comparison group (BCCG) will be defined by the eligible patients from facilities #9, #15, #1, #11, #23, and # 18 (refer to Table 3).

STEP 6.

Now we calculated the performance proportion of the “best-in-class” comparison group. The performance proportion of the “best-in-class” comparison group is known as the Pared Mean (PM).

The Pared Mean (PM) is the mean performance of the facilities above the break point (above the benchmark cut-off level).

The formula for the Pared Mean is

$$PM = (\text{Total number of eligible AMI patients receiving aspirin in the BCCG}) / (\text{Total number of eligible AMI patients in the BCCG}).$$

Therefore the Pared Mean (PM) = $79 / 87 = 0.908$.

NOTE: The raw data are used to calculate the Pared Mean (PM), not the APF!

STEP 7.

The Pared Mean Comparison.

In order to compare a facility's performance proportion to the "best-in-class" comparison group (BCCG), the facility's performance proportion needs to be calculated. It is calculated in a similar manner as the Pared Mean.

For sake of notation, we will refer to the facility's performance proportion as the unadjusted performance fraction (UPF).

The formula for the unadjusted performance fraction (UPF) is

$$\text{UPF} = (\text{Total number of eligible AMI patients receiving aspirin in the facility}) / (\text{Total number of eligible AMI patients in the facility}).$$

For facility # 1, $\text{UPF} = 29 / 33 = 0.879$.

NOTE: The raw data of the facility are used to calculate the unadjusted performance fraction (UPF).

The facility's unadjusted performance fraction (UPF) is compared to the "best-in-class" comparison group's performance fraction [the BCCG's Pared Mean (PM)].

When facility # 1 is compared to the "best-in-class" comparison group, we have the following:

Table 4. The BCCG Comparison of AMI-1: Aspirin at Arrival.

AMI-1: Aspirin at Arrival.

Facility UPF	BCCG PM
0.879	0.908

Standard comparison charts and graphs can be used with the UPF and BCCG PM proportions.

SAS® PROGRAM WITH SAS® MACRO

```

/*****
* Filename      : C:\SUGI 2006 paper\Best_in_Comparison.sas
* Contract      :
* Project       :
* Organization:  Maryland Medical Research Institute
* Authorship    : Nan Guo & William McCarthy
* Prod Date     : 09/2005
* Input         : C:\SUGI 2006 paper\csv files\SUGI2.csv
*               : Field Name:
*               : 1). facility: Facility #
*               : 2). dem      : # Eligible AMI Patients
*               : 3). x        : # Eligible AMI Patients Receiving
*               :               Aspirin
* Output        : \\rtf Files\Best_in_Class Comparison Group.rtf
*               : \\PDF Files\Graph_&dt7..emf
* Macro Lib     :
* Description    : BCCG Methodology
* Modification:
* Mod#         Date       Author/Description
*****/
options mprint symbolgen nocenter nonumber nodate
        pageno=1 ps=50 ls=180 orientation=landscape;

filename inpl 'C:\SUGI 2006 paper\csv files\SUGI2.csv'; * Table 1 raw data*;

```

```

title h=3 "SUGI 2006 paper - Best-in-Class Comparison Group";

data temp;
  length APF 5.3;
  infile inpfl dsd firstobs=2;
  input facility dem x;
  * --- to calculate Adjusted Performance Fraction value --- *;
  APF = round((x+1)/(dem+2),0.001);
  * --- writes first 10 obs values to the SAS log --- *;
  if _N_ le 10 then put _all_;
run;

* --- to count total eligible patients & total facilities--- *;
proc sql noprint;
  select sum(dem) as total,count(*) as cnts
  into :tot,:cnts
  from temp;
quit;

* --- sort data by descending order --- *;
proc sort data=temp
  out=sorted;
  by descending APF;
run;

* --- adds cumulative values into data set --- *;
data cumul_temp;
  retain demcuml xcuml ;
  set sorted;
  if _N_ eq 1 then
  do;
    demcuml = 0;
    xcuml = 0;
  end;
  demcuml = sum(demcuml,dem);
  xcuml = sum(xcuml,x);
  pct10 = round(&tot*0.10); /* It always top 10 percent */
  diff = demcuml - pct10;
  if _N_ le 10 then put _all_;
run;

proc sql noprint;
  select min(abs(demcuml-round(&tot*0.10))) as min
  into :closest
  from cumul_temp;
quit;

data culmu;
  set cumul_temp (where=(diff<= &closest)) end=EOF;
  if EOF then call symputx('mcrAPF',APF);
run;

data _null_;
  set cumul_temp(where=(APF >= input(symget('mcrAPF'),5.3))) end=EOF;
  if EOF then
  do;
    PM =round(xcuml/demcuml,.001); * Pared Mean *;
    tick=round(xcuml/demcuml,.1) ; * Pared Mean Line *;
    call symputx('mcrPM',PM) ; * assign PM value to a macro variable*;
    call symputx('mcrtick',tick*10+1);
  end;
run;

```

```

data final;
  set cumul_temp;
  UPF = x/dem;
  PM = input(symget('mcrPM'),8.7);
  label
    x      = 'Eligible AMI Patients Receiving Aspirin'
    dem    = 'Eligible AMI Patients'
    APF    = 'Adjusted Performance Fraction'
    PM     = 'Pared Mean'
    UPF    = 'Unadjusted Performance Fraction'
    facility = 'Facility #'
    demcuml = 'Cumulative # of Eligible AMI Patients'
    xcuml  = 'Cumulative # of Eligible AMI Patients Receiving Aspirin';
run;

%let sty=style(column)=[protectspecialchars=off cellwidth=0.85in just=1
pretext="\tqdec\tx550 "];
%let styl=style(column)=[protectspecialchars=off cellwidth=0.85in just=1
pretext="\tqdec\tx450 "];

proc template;
  define style Styles.Custom;
    parent = Styles.Printer;
    replace fonts /
      'TitleFont2' = ("Times Roman",8pt,Bold Italic)
      'TitleFont' = ("Times Roman",8pt,Bold Italic)
      'StrongFont' = ("Times Roman",8pt,Bold)
      'EmphasisFont' = ("Times Roman",8pt,Italic)
      'FixedEmphasisFont' = ("Courier",7pt,Italic)
      'FixedStrongFont' = ("Courier",7pt,Bold)
      'FixedHeadingFont' = ("Courier",7pt,Bold)
      'BatchFixedFont' = ("SAS Monospace, Courier",6.7pt)
      'FixedFont' = ("Courier",9pt)
      'headingEmphasisFont' = ("Times Roman",8pt,Bold Italic)
      'headingFont' = ("Times Roman",8pt,Bold)
      'docFont' = ("Times Roman",7pt)
      'FootFont' = ("Times Roman",7pt);

    replace Body from Document
      "Controls the Body file." /
      bottommargin = 0.05in
      topmargin = 0.05in
      rightmargin = 1.0in
      leftmargin = 1.0in
      foreground=red
      font_size=2;

    replace color_list
      "Colors used in the default style" /
      'link' = blue
      'bgH' = white
      'bgT' = white
      'bgD' = white
      'fg' = black
      'bg' = white;
  end;
run;

ods rtf close;
ods listing close;

```

```

title2 h=3 "Best-in-Class Comparison Group Pared Mean (BCCG PM): &mcrPM";

footnotel italic bold h=2 j=1 "Pgm:\\Cre_Best_In_Class_Comparison_Group.sas"
      j=r "Run:&sysdate";

ods rtf file="C:\SUGI 2006 paper\rtf Files\Best_in_Class Comparison Group.rtf"
style=Custom;

proc report data=final nowd;
  column facility dem x demcuml xcuml APF UPF;
  define facility / display 'Facility #' &sty;
  define dem      / display 'Eligible AMI Patients' &sty;
  define x        / display 'Eligible AMI Patients Receiving Aspirin' &sty;
  define APF      / display 'Adjusted Performance Fraction' format=7.3 &styl;
  define UPF      / display 'Unadjusted Performance Fraction' format=7.3 &styl;
  define demcuml  / display 'Cumulative # of Eligible AMI Patients' &sty;
  define xcuml    / display 'Cumulative # of Eligible AMI Patients Receiving Aspirin'
&sty;
run;

ods rtf close;
ods listing;

%macro fdate(fmt);
  %global fdate dt7;
  data _null_;
    call symput("fdate",left(put("&sysdate"d,&fmt)));
    call symputx("dt7",put(date(),date7.));
  run;
%mend fdate;

%fdate(mmddy10.)

goption reset=all vpos=300 hpos=200 device=emf gsfname=GSASFILE ftext='Arial' htext=3
gunit=pct display rotate=landscape;

FILENAME GSASFILE "C:\SUGI 2006 paper\PDF Files\Graph_&dt7..emf";

title2 height=2.8 "Best-in-Class Comparison Group Pared Mean (BCCG PM):" color=red
height=3.0 "&mcrPM";

footnotel j=1 " Pgm:\\Cre_Graph.sas" j=r "Run Date: &fdate";

symbol1 l=1 c=red v=circle height=3 width=5;

axis1 label=('Facility #') value=(height=3.0) minor=(number=1) order=(1 to
&cnts by 1) offset=(3,3);
axis2 minor=none
order=(0 to 1.2 by 0.1)
value=( tick=&mrtick color=red height=3.0) offset=(,3)
label=(angle=90 'Unadjusted Performance Fraction' height=0.2) ;
legend1 frame
label=none
value=(height=1.3 'PM' 'APF')
across=1
position=(top left inside)
shape=symbol(6,1.5)
offset=(0 cm, 0.0010 cm)
mode=share;

```



```

proc gplot data=final;
  plot UPF*facility/vaxis=axis2
      haxis=axis1
      vref=&mcrPM
      cvref=blue
      lvref=2
      legend=legend1
      frame;


```

```
run;
```

```
quit;
```

SAS® OUTPUT

Facility #	Eligible AMI Patients	Eligible AMI Patients Receiving Aspirin	Cumulative # of Eligible AMI Patients	Cumulative # of Eligible AMI Patients Receiving Aspirin	Adjusted Performance Fraction	Unadjusted Performance Fraction
9	7	7	7	7	0.889	1.000
15	23	21	30	28	0.880	0.913
1	33	29	63	57	0.857	0.879
11	16	14	79	71	0.833	0.875
23	4	4	83	75	0.833	1.000
18	4	4	87	79	0.833	1.000
8	33	28	120	107	0.829	0.848
16	31	26	151	133	0.818	0.839
13	8	7	159	140	0.800	0.875
14	28	23	187	163	0.800	0.821
20	3	3	190	166	0.800	1.000
6	56	45	246	211	0.793	0.804
5	123	95	369	306	0.768	0.772
21	2	2	371	308	0.750	1.000
17	62	46	433	354	0.734	0.742
12	28	21	461	375	0.733	0.750
22	85	61	546	436	0.713	0.718
19	15	11	561	447	0.706	0.733
2	1	1	562	448	0.667	1.000
24	65	41	627	489	0.627	0.631
3	0	0	627	489	0.500	.
10	15	7	642	496	0.471	0.467
7	64	29	706	525	0.455	0.453
4	1	0	707	525	0.333	0.000



Pgm:\Cre_Graph.sas

Run Date: 11/08/2005

REFERENCES

1. Based on the Achievable Benchmarks of Care (ABC™) System developed by the Center for Outcomes and Effectiveness Research and Education (COERE), University of Alabama at Birmingham.
2. Agresti, A. (1990). Categorical Data Analysis. John Wiley & Sons, NY (pp463-464).

CONTACT INFORMATION

William F. McCarthy
Principal Statistician
Maryland Medical Research Institute
600 Wyndhurst Avenue
Baltimore, Maryland 21210-2425
(410) 435-4200
wmccarthy@mmri.org

Nan Guo
Certified Advanced SAS Programmer
Maryland Medical Research Institute
600 Wyndhurst Avenue
Baltimore, Maryland 21210-2425
(410) 435-4200
nguo@mmri.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.